<commit_before>{lib, stdenv, fetchurl}:
stdenv.mkDerivation rec {
  pname = "fairymax";
  version = "4.8";
  src = fetchurl {
    url = "http://home.hccnet.nl/h.g.muller/fmax4_8w.c";
    sha256 = "01vxhpa4z0613mkgkzmsln293wxmyp5kdzil93cd1ik51q4gwjca";
  };
  ini = fetchurl {
    url = "http://home.hccnet.nl/h.g.muller/fmax.ini";
    sha256 = "1zwx70g3gspbqx1zf9gm1may8jrli9idalvskxbdg33qgjys47cn";
  };
  unpackPhase = ''
    cp ${src} fairymax.c
    cp ${ini} fmax.ini
  '';
  buildPhase = ''
    $CC *.c -Wno-return-type -o fairymax -DINI_FILE='"'"$out/share/fairymax/fmax.ini"'"'
  '';
  installPhase = ''
    mkdir -p "$out"/{bin,share/fairymax}
    cp fairymax "$out/bin"
    cp fmax.ini "$out/share/fairymax"
  '';
  meta = {
    description = "A small chess engine supporting fairy pieces";
    longDescription = ''
       A version of micro-Max that reads the piece description
       from a file fmax.ini, so that arbitrary fairy pieces can be
       implemented. This version (4.8J) supports up to 15 piece types,
       and board sizes up to 12x8. A Linux port exists in the
       format of a debian package.
    '';
    license = lib.licenses.free ;
    maintainers = [lib.maintainers.raskin];
    platforms = lib.platforms.all;
    homepage = "http://home.hccnet.nl/h.g.muller/dwnldpage.html";
  };
}<commit_msg>fairymax: use SRI hash format<commit_after>{ lib
, stdenv
, fetchurl
}:

stdenv.mkDerivation rec {
  pname = "fairymax";
  version = "4.8";

  src = fetchurl {
    url = "http://home.hccnet.nl/h.g.muller/fmax4_8w.c";
    hash = "sha256-ikn+CA5lxtDYSDT+Nsv1tfORhKW6/vlmHcGAT9SFfQc=";
  };

  ini = fetchurl {
    url = "http://home.hccnet.nl/h.g.muller/fmax.ini";
    hash = "sha256-lh2ivXx4jNdWn3pT1WKKNEvkVQ31JfdDx+vqNx44nf8=";
  };

  unpackPhase = ''
    cp ${src} fairymax.c
    cp ${ini} fmax.ini
  '';

  buildPhase = ''
    cc *.c -Wno-return-type \
      -o fairymax \
      -DINI_FILE='"'"$out/share/fairymax/fmax.ini"'"'
  '';

  installPhase = ''
    mkdir -p "$out"/{bin,share/fairymax}
    cp fairymax "$out/bin"
    cp fmax.ini "$out/share/fairymax"
  '';

  meta = with lib; {
    homepage = "http://home.hccnet.nl/h.g.muller/dwnldpage.html";
    description = "A small chess engine supporting fairy pieces";
    longDescription = ''
       A version of micro-Max that reads the piece description from a file
       fmax.ini, so that arbitrary fairy pieces can be implemented. This version
       (4.8J) supports up to 15 piece types, and board sizes up to 12x8.
    '';
    license = licenses.free;
    maintainers = [ maintainers.raskin ];
    platforms = platforms.all;
  };
}
{'fraction_non_alphanumeric': 0.14225781845919147, 'fraction_numerical': 0.02936689549961861, 'mean_word_length': 2.8630338733431517, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4915482', 'n_tokens_mistral': 1091, 'n_tokens_neox': 1039, 'n_words': 216}